Steel and Composite Structures, *Vol. 16, No. 3 (2014) 233-252*
DOI: http://dx.doi.org/10.12989/scs.2014.16.3.233

Automatic analysis of thin-walled laminated composite sections

A. Prokić [*1], D. Lukić [1] and Dj. Ladjinović [2]

[1] *Faculty of Civil Engineering, University of Novi Sad, Kozaračka 2a, 24000 Subotica, Serbia*
[2] *Faculty of Tehnical Sciences, University of Novi Sad, Trg Dositeja Obradovica, 21000 Novi Sad, Serbia*

(*Received June 09, 2012, Revised October 16, 2013, Accepted October 30, 2013*)

Abstract. In this paper a computer program is developed for the determination of geometrical and material properties of composite thin-walled beams with arbitrary open cross-section and any arbitrary laminate stacking sequence. Theory of thin-walled composite beams is based on assumptions consistent with the Vlasov's beam theory and classical lamination theory. The program is written in Fortran 77. Some numerical examples are given, with complete information about input and output.

Keywords: thin-walled composite beam; open section; computer program; classical lamination theory; arbitrary lamination

1. Introduction

Thin-walled composite elements has recently become the focus of intense researches as a result of their expanded use as structural components within the fields of mechanical, civil, aeronautical engineering, and other industries. These structural components made of advanced composite materials are ideal for structural applications because of the high strength-to-weight and stiffness-to-weight ratios. Another advantage of composites is their flexibility in design. For example, mechanical properties of the laminate can be altered simply by changing the stacking sequence, fibre lay-up and thickness of each ply. Consequently, design may be optimized under different set of conditions to achieve the optimal performance of the structure.

In structural analysis it is often necessary to determine the material-geometry properties of thin-walled composite beams, with open cross-sections. The hand calculation used to determine them, although mostly elementary, are tedious, time consuming and numerical errors are easily introduced. Though many papers are written on behavior of thin-walled composite beams (Banerjee 1998, Banerjee and Su 2006, Cardoso *et al.* 2009, Cardoso and Valido 2011, Chen and Hsiao 2007, Kim *et al.* 2007 and 2008, Lee 2001, Machado and Cortınez 2005, Mechado *et al.* 2007, Piovan and Cortınez 2007, Rajasekaran 2005, Sapountzakis and Tsiatas 2007, Sapountzakis and Mokos 2007, Vo and Lee 2009, Vo *et al.* 2011), to the authors' knowledge, no general computer program for the determination of the material-geometry properties of thin laminated section is available. The computer program presented here is fairly common and gives the structural designer the ability to analyze thin-walled composite sections of any shape and arbitrary

*Corresponding author, Professor, E-mail: aprokic@EUnet.rs

http://www.techno-press.org/?journal=scs&subpage=8 ISSN: 1229-9367 (Print), 1598-6233 (Online)



Fig. 1 Geometry and reference systems

laminate stacking sequence quickly and efficiently. This paper is an extension of the author's previous works (Prokić 1999 and 2000).

The number of input data is minimized and there are no set rules to follow in the joint, element and lamina numbering, which makes program's application easy even by a practing engineer who cannot go into the details of composite thin-walled theory.

2. Basic theory

A straight thin-walled laminated composite beam of length l with an open cross-section is considered (Fig. 1). In order to determine the geometry of the cross-section of the beam two coordinate systems are used. The first of these is an orthogonal Cartesian coordinate systems (x, y, z) for which the z-axis is parallel to the longitudinal axis of the beam. The second coordinate system is a local one (e, s, z) where e and s are profile coordinates measured along the normal to the contour (the midline of the cross-section) and along the contour line, respectively. The (e, s, z) and (x, y, z) coordinate systems are related through an angle of orientation α. The coordinates of the contour in the (x, y, z) coordinate system are ($\bar{x}$, $\bar{y}$, z). Point P is called the pole.

2.1 Kinematics of the beam

Following Vlasov's beam theory the basic assumptions of thin-walled laminated beams are introduced.

- The cross-section of the beam is not distorted during the deformation of the beam.
- The shear strains in the middle surface of the wall are negligible.
- The Kirchhoff-Love assumption in classical plate theory remains valid for laminated composite thin-walled beams.

Based on assumption above, the displacements u_*, v_* and w_* at any point on the beam cross-section can be expressed by four components, two translations u_P, v_P of arbitrarily taken pole P, the cross-section rotation φ about the pole P, and axial displacement w of centoid.

$$\begin{aligned} u_* &= u_P - \left(y - y_P\right)\varphi \\ v_* &= v_P + \left(x - x_P\right)\varphi \\ w_* &= w - u'_P x - v'_P y - \varphi' \omega_P \end{aligned} \tag{1}$$

where ω_p

$$\omega_P = \int_0^s h_P ds + h_{nP} e \tag{2}$$

is generalized warping function with respect to pole P.

h_P and h_{nP}, perpendicular distance from tangent and normal at arbitrary point of cross-section to the point P, are positive when normal $\vec{n}$ and tangent $\vec{t}$, respectively, are rotating counterclockwise about the pole P, when observed from the positive z direction. The second term on the right-hand side of Eq. (2) determines the relative warping in relation to the midline of cross-section. This term has little effect on the torsional properties of a thin profile and most frequently is neglected in the technical theory of thin-walled beams. However, its inclusion does not present additional difficulties, and therefore this term has been included in the computer program. The warping by definition must be the same at a node where a number of members are joined together.

Consistent with displacement field, Eq. (1), the non-vanishing strain components are

$$\begin{aligned} \varepsilon_z &= w' - u''_P x - v''_P y - \varphi'' \omega_P \\ \gamma_s &= 2\varphi' e \end{aligned} \tag{3}$$

2.2 Constitutive equations

For a unidirectionally reinforced lamina the stress-strain relations is

$$\begin{bmatrix} \sigma_1 \\ \sigma_2 \\ \tau_{12} \end{bmatrix}_k = \begin{bmatrix} Q_{11} & Q_{12} & \\ Q_{12} & Q_{11} & \\ & & Q_{66} \end{bmatrix}_k \begin{bmatrix} \varepsilon_1 \\ \varepsilon_2 \\ \gamma_{12} \end{bmatrix} \tag{4}$$

where the terms Q_{ij} are so-called reduced stiffnesses Jones (1975) for a plane stress state in the 1-2 plane of lamina k. The terms Q_{ij} are made up of material property with respect to each layer and can be shown in terms of the engineering constants

$$Q_{11} = \frac{E_1}{1 - \nu_{12}^2 \dfrac{E_2}{E_1}} \qquad Q_{12} = \frac{\nu_{12} E_2}{1 - \nu_{12}^2 \dfrac{E_2}{E_1}} \qquad Q_{22} = \frac{E_2}{1 - \nu_{12}^2 \dfrac{E_2}{E_1}} \qquad Q_{66} = G_{12} \tag{5}$$

Generally, the principal material coordinates 1-2 for orthotropic lamina k do not coincide with beam coordinates s-z. If the principal 1-axis making an angle ϑ with respect to reference z-axis the stress-strain relation in s-z coordinate system is

$$\begin{bmatrix} \sigma_z \\ \sigma_s \\ \tau_{sz} \end{bmatrix}_k = \begin{bmatrix} \bar{Q}_{11} & \bar{Q}_{12} & \bar{Q}_{16} \\ \bar{Q}_{12} & \bar{Q}_{22} & \bar{Q}_{26} \\ \bar{Q}_{16} & \bar{Q}_{26} & \bar{Q}_{66} \end{bmatrix}_k \begin{bmatrix} \varepsilon_z \\ \varepsilon_s \\ \gamma_{sz} \end{bmatrix} \tag{6}$$

in which transformed reduced stiffness $\bar{Q}_{ij}$ are

$$\begin{aligned}
\bar{Q}_{11} &= Q_{11}m^4 + Q_{22}n^4 + 2m^2n^2\left(Q_{12} + 2Q_{66}\right) \\
\bar{Q}_{12} &= m^2n^2\left(Q_{11} + Q_{22} - 4Q_{66}\right) + \left(m^4 + n^4\right)Q_{12} \\
\bar{Q}_{16} &= \left[Q_{11}m^2 - Q_{22}n^2 - \left(Q_{12} + 2Q_{66}\right)\left(m^2 - n^2\right)\right]mn \\
\bar{Q}_{22} &= Q_{11}n^4 + Q_{22}m^4 + 2m^2n^2\left(Q_{12} + 2Q_{66}\right) \\
\bar{Q}_{26} &= \left[Q_{11}n^2 - Q_{22}m^2 + \left(Q_{12} + 2Q_{66}\right)\left(m^2 - n^2\right)\right]mn \\
\bar{Q}_{66} &= \left(Q_{11} + Q_{22} - 2Q_{12}\right)m^2n^2 + Q_{66}\left(m^2 - n^2\right)^2
\end{aligned} \tag{7}$$

where $m = \cos\vartheta$ and $n = \sin\vartheta$.

By using free stress in contour direction, $\sigma_s = 0$, the above equation can be simplified as

$$\begin{bmatrix} \sigma_z \\ \tau_s \end{bmatrix}_k = \begin{bmatrix} \bar{\bar{Q}}_{11} & \bar{\bar{Q}}_{16} \\ \bar{\bar{Q}}_{16} & \bar{\bar{Q}}_{66} \end{bmatrix}_k \begin{bmatrix} \varepsilon_z \\ \gamma_s \end{bmatrix} \tag{8}$$

where

$$\bar{\bar{Q}}_{11} = \bar{Q}_{11} - \frac{\bar{Q}_{12}^2}{\bar{Q}_{22}} \qquad \bar{\bar{Q}}_{16} = \bar{Q}_{16} - \frac{\bar{Q}_{12}\bar{Q}_{26}}{\bar{Q}_{22}} \qquad \bar{\bar{Q}}_{66} = \bar{Q}_{66} - \frac{\bar{Q}_{26}^2}{\bar{Q}_{22}} \tag{9}$$

Stress resultants at the cross-section can be derived by integrating the corresponding stresses over the cross sectional area, as given by

$$\begin{aligned}
N &= \iint_F \sigma_z dF \\
M_x &= \iint_F \sigma_z y dF \\
M_y &= -\iint_F \sigma_z x dF \\
M_{\omega P} &= \iint_F \sigma_z \omega_P dF \\
T_s &= 2\iint_F \tau_s e dF
\end{aligned} \tag{10}$$

In Eq. (10), N represents the axial force, M_x and M_y the bending moments with respect to the x and y axis, T_s the Saint Venant torque, $M_{\omega P}$ the bimoment and F the area of the cross-section. Taking into account the Eqs. (3) and (8) the forces may be defined in terms of componential displacements as

$$
\begin{aligned}
N &= \iint_F \left[\bar{\bar{Q}}_{11}\left(w' - u_P''x - v_P''y - \varphi''\omega_P\right) + \bar{\bar{Q}}_{16} 2\varphi' e \right] dF \\
M_x &= \iint_F \left[\bar{\bar{Q}}_{11}\left(w' - u_P''x - v_P''y - \varphi''\omega_P\right) y + \bar{\bar{Q}}_{16} 2\varphi' e y \right] dF \\
M_y &= -\iint_F \left[\bar{\bar{Q}}_{11}\left(w' - u_P''x - v_P''y - \varphi''\omega_P\right) x + \bar{\bar{Q}}_{16} 2\varphi' e x \right] dF \\
M_{\omega P} &= \iint_F \left[\bar{\bar{Q}}_{11}\left(w' - u_P''x - v_P''y - \varphi''\omega_P\right) \omega_P + \bar{\bar{Q}}_{16} 2\varphi' e \omega_P \right] dF \\
T_s &= 2\iint_F \left[\bar{\bar{Q}}_{16}\left(w' - u_P''x - v_P''y - \varphi''\omega_P\right) e + \bar{\bar{Q}}_{66} 2\varphi' e^2 \right] dF
\end{aligned} \tag{11}
$$

or, written in matrix form

$$
\begin{bmatrix} N \\ M_y \\ -M_x \\ -M_{\omega P} \\ T_s \end{bmatrix} =
\begin{bmatrix}
A & -S_x & -S_y & -S_{\omega P} & S_e \\
-S_x & I_{xx} & I_{xy} & I_{x\omega P} & -I_{xe} \\
-S_y & I_{xy} & I_{yy} & I_{y\omega P} & -I_{ye} \\
-S_{\omega P} & I_{x\omega P} & I_{y\omega P} & I_{\omega P\omega P} & -I_{\omega Pe} \\
S_e & -I_{xe} & -I_{ye} & -I_{\omega Pe} & I_{ee}
\end{bmatrix}
\begin{bmatrix} w' \\ u_P'' \\ v_P'' \\ \varphi'' \\ \varphi' \end{bmatrix} \tag{12}
$$

in which

$$
\begin{aligned}
A &= \iint_F \bar{\bar{Q}}_{11} dF = \int_s A_{11} ds \\
S_x &= \iint_F \bar{\bar{Q}}_{11} x dF = \int_s \left(A_{11}\bar{x} + B_{11}\cos\alpha\right) ds \\
S_y &= \iint_F \bar{\bar{Q}}_{11} y dF = \int_s \left(A_{11}\bar{y} + B_{11}\sin\alpha\right) ds \\
S_{\omega P} &= \iint_F \bar{\bar{Q}}_{11} \omega_P dF = \int_s \left(A_{11}\bar{\omega}_P + B_{11}h_{nP}\right) ds \\
I_{xx} &= \iint_F \bar{\bar{Q}}_{11} x^2 dF = \int_s \left(A_{11}\bar{x}^2 + 2B_{11}\bar{x}\cos\alpha + D_{11}\cos^2\alpha\right) ds \\
I_{yy} &= \iint_F \bar{\bar{Q}}_{11} y^2 dF = \int_s \left(A_{11}\bar{y}^2 + 2B_{11}\bar{y}\sin\alpha + D_{11}\sin^2\alpha\right) ds \\
I_{xy} &= \iint_F \bar{\bar{Q}}_{11} xy dF = \int_s \left[A_{11}\bar{x}\,\bar{y} + B_{11}\left(\bar{x}\sin\alpha + \bar{y}\cos\alpha\right) + D_{11}\sin\alpha\cos\alpha\right] ds \\
I_{x\omega P} &= \iint_F \bar{\bar{Q}}_{11} \omega_P x dF = \int_s \left[A_{11}\bar{x}\,\bar{\omega}_P + B_{11}\left(\bar{x}h_{nP} + \bar{\omega}_P\cos\alpha\right) + D_{11}h_{nP}\cos\alpha\right] ds \\
I_{y\omega P} &= \iint_F \bar{\bar{Q}}_{11} \omega_P y dF = \int_s \left[A_{11}\bar{y}\,\bar{\omega}_P + B_{11}\left(\bar{y}h_{nP} + \bar{\omega}_P\sin\alpha\right) + D_{11}h_{nP}\sin\alpha\right] ds \\
I_{\omega P\omega P} &= \iint_F \bar{\bar{Q}}_{11} \omega_P^2 dF = \int_s \left(A_{11}\bar{\omega}_P^2 + 2B_{11}\bar{\omega}_P h_{nP} + D_{11}h_{nP}^2\right) ds \\
S_e &= 2\iint_F \bar{\bar{Q}}_{16} e dF = 2\int_s B_{16} ds \\
&\downarrow
\end{aligned} \tag{13}
$$

$$
\begin{aligned}
I_{xe} &= 2\iint_F \bar{\bar{Q}}_{16} x e\, dF = 2\int_s \left(B_{16}\bar{x} + D_{16}\cos\alpha\right) ds \\
I_{ye} &= 2\iint_F \bar{\bar{Q}}_{16} y e\, dF = 2\int_s \left(B_{16}\bar{y} + D_{16}\sin\alpha\right) ds \\
I_{e\omega P} &= 2\iint_F \bar{\bar{Q}}_{16} \omega_P e\, dF = 2\int_s \left(B_{16}\bar{\omega}_P + D_{16}h_{nP}\right) ds \\
I_{ee} &= 4\iint_F \bar{\bar{Q}}_{66} e^2 dF = 4\int_s D_{66} ds
\end{aligned}
\tag{13}
$$

where

$$
\begin{aligned}
x &= \bar{x} + e\cos\alpha \\
y &= \bar{y} + e\sin\alpha \\
\omega_P &= \bar{\omega}_P + h_{nP} e
\end{aligned}
\tag{14}
$$

and

$$
\begin{aligned}
A_{ij} &= \int \bar{\bar{Q}}_{ij} de \\
B_{ij} &= \int \bar{\bar{Q}}_{ij} e\, de \\
D_{ij} &= \int \bar{\bar{Q}}_{ij} e^2 de
\end{aligned}
\tag{15}
$$

2.3 Center of gravity and shear center

In all integrals (13) are incorporated both the geometry and material properties of cross-section. By appropriate selection of Cartesian coordinate system, pole P and starting point O_1 we can achieve that

$$
S_x = S_y = I_{xz} = I_{\omega P} = I_{x\omega P} = I_{y\omega P} = 0 \tag{16}
$$

So, we get the simplified expressions for stress resultants

$$
\begin{bmatrix} N \\ M_y \\ -M_x \\ -M_{\omega P} \\ T_s \end{bmatrix}
=
\begin{bmatrix}
A & 0 & 0 & 0 & S_e \\
0 & I_{xx} & 0 & 0 & -I_{xe} \\
0 & 0 & I_{yy} & 0 & -I_{ye} \\
0 & 0 & 0 & I_{\omega p\omega P} & -I_{\omega pe} \\
S_e & -I_{xe} & -I_{ye} & -I_{\omega pe} & I_{ee}
\end{bmatrix}
\begin{bmatrix} w' \\ u_P'' \\ v_P'' \\ \varphi'' \\ \varphi' \end{bmatrix}
\tag{17}
$$

In this case, using the principle of virtual work Prokić (1996), the governing equations of thin walled composite beam can be written with displacements as primary unknowns

$$
\begin{aligned}
&A w'' + S_e \varphi'' = -p_z \\
&I_{xx} u_P'''' - I_{xe}\varphi''' = p_x - m_y' \\
&I_{yy} v_P'''' - I_{ye}\varphi''' = p_y + m_x' \\
&I_{\omega_D\omega_D}\varphi'''' - S_e w'' + I_{xe} u_P''' + I_{ye} v_P''' - I_{ee}\varphi'' = m_P + m_{\omega_P}'
\end{aligned}
\tag{18}
$$

Analyzing the system of equations above, we can conclude that the point of cross-section P, which satisfy conditions $I_{\omega P} = I_{x\omega P} = I_{y\omega P} = 0$, does not have the same significance as in the classical theory of thin-walled beams (shear center). Torsion and bending in this case cannot be separated and are coupled together with the extension.

The standard procedure for evaluating the 'center of gravity' and 'shear center' of open profile was described by Murray (1984). Only a brief recapitulation of procedure will be given at this point.

- In the first step we find all the section properties starting with an arbitrary set of axes xOy, with pole P located at the origin and starting point O_1 located at the first joint of profile.
- A parallel shift of reference axes to the point C , whose coordinates are defined as

$$x_c = \frac{S_x}{F} \qquad y_c = \frac{S_y}{F} \tag{19}$$

The location of P is retained but the coordinates of the starting point are chosen in a way which makes

$$S_\omega = 0 \tag{20}$$

- The axes x and y should be rotated in the direction of the principal axes x and y, the angle of rotation ψ being given by

$$\tan 2\psi = \frac{2I_{xy}}{I_{xx} - I_{yy}} \tag{21}$$

The pole is moved to the point which is chosen so that

$$I_{x\omega P} = 0 \qquad I_{y\omega P} = 0 \tag{22}$$

3. Numerical procedure

The arbitrary midline of the cross-section is approximated by a polygonal one. In this case the section is composed of a series of mutually connected prismatic thin-walled elements (segments). The number of elements adopted depends on the desired accuracy. Points at which two or more elements are connected will be indicated as joints of cross-section. Joints and elements may be marked arbitrary, and any of the end joints of en element may be chosen as the first joint.

Each orthotropic layer of laminate is defined by its thickness, its location in the laminate, its material properties and fibers orientation. Note that the contour coordinate s is oriented from the initial node to the final node of the element, which affects the n-axis orientation. The fiber orientation of layer k is given by the angle ϑ_k which is positive counterclockwise around n-axis and starting from the z-axis.

Marking the joints of an element with i and k, (Fig. 2), we may write

$$\begin{aligned} x &= x_i - s\sin\alpha + e\cos\alpha \\ y &= y_i + s\cos\alpha + e\sin\alpha \end{aligned} \tag{23}$$

and further, supposing that the pole P is located at the origin



Fig. 2 Polygonal segment showing the definition of the various variables

$$\begin{aligned} \omega_P &= \bar{\omega}_P + h_{nP}e = \omega_{Pi} - \left(\omega_{Pi} - \omega_{Pk}\right)\frac{s}{l} + h_{nP}e \\ h_{nP} &= x_i \sin\alpha - y_i \cos\alpha - s \\ h_{nP(i)} &= x_i \sin\alpha - y_i \cos\alpha \end{aligned} \tag{24}$$

Now, all integrals (13) can be derived explicitly in the form convenient for programming, Prokić (1996)

$$\begin{aligned} A &= \sum_{ns} A_{11}\, l \\ S_x &= \sum_{ns}\left(A_{11}\frac{x_i + x_k}{2} + B_{11}\cos\alpha \right) l \\ S_y &= \sum_{ns}\left(A_{11}\frac{y_i + y_k}{2} + B_{11}\sin\alpha \right) l \\ I_{xx} &= \sum_{ns}\left[A_{11}\frac{x_i^2 + x_k^2 + x_i x_k}{3} + B_{11}\left(x_i + x_k\right)\cos\alpha + D_{11}\cos^2\alpha \right] l \\ I_{yy} &= \sum_{ns}\left[A_{11}\frac{y_i^2 + y_k^2 + y_i y_k}{3} + B_{11}\left(y_i + y_k\right)\sin\alpha + D_{11}\sin^2\alpha \right] l \\ I_{xy} &= \sum_{ns}\left[A_{11}\frac{2x_i y_i + 2x_k y_k + x_i y_k + x_k y_i}{6} + B_{11}\left(\frac{x_i + x_k}{2}\sin\alpha + \frac{y_i + y_k}{2}\cos\alpha \right) + D_{11}\sin\alpha\cos\alpha \right] l \\ S_{\omega P} &= \sum_{ns}\left[A_{11}\frac{\omega_{Pi} + \omega_{Pk}}{2} + B_{11}\left(h_{nP(i)} - \frac{1}{2}l \right) \right] l \end{aligned} \tag{25}$$

↓

↑

$$I_{x\omega p}=\sum_{ns}\left[A_{11}\frac{2x_i\omega_{Pi}+2x_k\omega_{Pk}+x_i\omega_{Pk}+x_k\omega_{Pi}}{6}+B_{11}\left(\frac{\omega_{Pi}+\omega_{Pk}}{2}\cos\alpha-l\frac{x_i+2x_k}{6}+\frac{x_i+x_k}{2}h_{nP(i)}\right)\right.$$
$$\left.+D_{11}\left(h_{nP(i)}\cos\alpha\text{-}\frac{1}{2}l\cos\alpha\right)\right]l \quad (2$$

$$I_{y\omega p}=\sum_{ns}\left[A_{11}\frac{2y_i\omega_{Pi}+2y_k\omega_{Pk}+y_i\omega_{Pk}+y_k\omega_{Pi}}{6}+B_{11}\left(\frac{\omega_{Pi}+\omega_{Pk}}{2}\sin\alpha-l\frac{y_i+2y_k}{6}+\frac{y_i+y_k}{2}h_{nP(i)}\right)\right.$$
$$\left.+D_{11}\left(h_{nP(i)}\sin\alpha-\frac{1}{2}l\sin\alpha\right)\right]l$$

$$I_{\omega p\omega p}=\sum_{ns}\left\{A_{11}\frac{\omega_{Pi}^2+\omega_{Pk}^2+\omega_{Pi}\omega_{Pk}}{3}+B_{11}\left[\left(\omega_{Pi}+\omega_{Pk}\right)h_{np(i)}-l\frac{\omega_{Pi}+2\omega_{Pk}}{3}\right]+D_{11}\left(h_{nP(i)}^2-h_{np(i)}l+\frac{1}{3}l^2\right)\right\}l \quad (25)$$

$$S_e=2\sum_{ns}B_{16}l$$

$$I_{xe}=2\sum_{ns}\left(B_{16}\frac{x_i+x_k}{2}+D_{16}\cos\alpha\right)l$$

$$I_{ye}=2\sum_{ns}\left(B_{16}\frac{y_i+y_k}{2}+D_{16}\sin\alpha\right)l$$

$$I_{\omega pe}=2\sum_{ns}\left[B_{16}\frac{\omega_{Pi}+\omega_{Pk}}{2}+D_{16}\left(h_{nP(i)}-\frac{1}{2}l\right)\right]l$$

$$I_{ee}=4\sum_{ns}D_{66}l$$

Σ represents the sum of each segment of the cross-section, and

$$A_{ij}=\sum_{k=1}^{nl}\overline{\overline{Q}}_{ij,k}\left(h_k-h_{k-1}\right)=\sum_{k=1}^{nl}\overline{\overline{Q}}_{ij,k}t_k$$
$$B_{ij}=\frac{1}{2}\sum_{k=1}^{nl}\overline{\overline{Q}}_{ij,k}\left(h_k^2-h_{k-1}^2\right)=\sum_{k=1}^{nl}\overline{\overline{Q}}_{ij,k}e_{T,k}t_k \quad (26)$$
$$D_{ij}=\frac{1}{3}\sum_{k=1}^{nl}\overline{\overline{Q}}_{ij,k}\left(h_k^3-h_{k-1}^3\right)=\sum_{k=1}^{nl}\overline{\overline{Q}}_{ij,k}\left(e_{T,k}^2t_k+\frac{t_k^3}{12}\right)$$

4. Computer program

NUME = number of A computer program in FORTRAN 77 is developed, capable of analyzing both the geometry and the material properties of thin-walled composite beams with arbitrary open cross-section. A following data should be prepared according to their respective format, and in the order in which they should be entered:

SET 1 (A20)
UFILE = name of file with input data

SET 2 (3I5)
NUMJ = number of joints elements
NUMM = number of different materials

SET 3 (4F10.0)
E1(I) = Young's moduli in the 1-direction for material of type (I)
E2(I) = Young's moduli in the 2-direction for material of type (I)
P12(I) = Poisson's ratio for material of type (I)
G12(I) = shear moduli in the 1-2 plane for material of type (I)

SET 4 (2F10.0)
X(I) = x-coordinate of joint "I", with reference to arbitrary chosen set of axes
Y(I) = y-coordinate of joint "I", with reference to arbitrary chosen set of axes

SET 5 (3I5)
N1(I) = first joint number of element "I"
N2(I) = second joint number of element "I"
NL = number of layers (laminas) for element "I"

SET 6 (3F10.0, I5)
TL = thickness of the layer (J) of element (I)
EL =distance to the centroid of layer (J) of element (I)
OL = angle of orientation of layer (J) of element (I)
TM = type of material (J) of element (I)

A listing of the program is given in Appendix A.

5. Illustrative examples

A thin-walled composite beam with channel cross-section shown in Fig. 3, is considered, (Cardoso *et al*. 2009). The cross-section consists of three equal laminates (elements), identified in the figure as 1, 2 and 3, each of them with four layers $[45/-45]_s$ and total thickness $t = 3$ mm.

The following engineering constants of composite beam, corresponding to S2-glass/epoxy, are used

$$\begin{aligned} E_1 &= 48.3 \text{ GPa} \\ E_2 &= 19.8 \text{ GPa} \\ \nu_{12} &= 0.27 \\ G_{12} &= 8.96 \text{ GPa} \end{aligned} \tag{27}$$

The coordinate system is established arbitrary and the elements and joints are numbered as shown. The description of input data and computer output is given in Tables 1 and 2.



Fig. 3 Channel cross-section

Table 1 Input data

4	3	1		
	48.3	19.8	0.27	8.96
	0.	0.		
	50.	50.		
	50.	0.		
	0.	50.		
2	4	4		
	0.75	1.125	0.785398	1
	0.75	0.375	-0.785398	1
	0.75	-0.375	-0.785398	1
	0.75	-1.125	0.785398	1
3	1	4		
	0.75	12125	0.785398	1
	0.75	0.375	-0.785398	1
	0.75	-0.375	-0.785398	1
	0.75	-1.125	0.785398	1
2	3	4		
	0.75	1.125	0.785398	1
	0.75	0.375	-0.785398	1
	0.75	-0.375	-0.785398	1
	0.75	-1.125	0.785398	1

Table 2 Output data

Number of Joints = 4
Number of elements = 3
Number of materials = 1

Type of material	E1	E2	N1	G
1	0.48300E+02	.19800E+02	.27000E+00	.89600E+01

Element 1				
Lamina	Thickness	Dist. to the cent. Line	Angle of orient.	Type of mater.
1	.75000E+00	.11250E+01	.78540E+00	1
2	.75000E+00	.37500E+00	-.78540E+00	1
3	.75000E+00	-.37500E+00	-.78540E+00	1
4	.75000E+00	-.11250E+01	.78540E+00	1

Element 2				
Lamina	Thickness	Dist. to the cent. Line	Angle of orient.	Type of mater.
1	.75000E+00	.11250E+01	.78540E+00	1
2	.75000E+00	.37500E+00	-.78540E+00	1
3	.75000E+00	-.37500E+00	-.78540E+00	1
4	.75000E+00	-.11250E+01	.78540E+00	1

Element 3				
Lamina	Thickness	Dist. to the cent. Line	Angle of orient.	Type of mater.
1	.75000E+00	.11250E+01	.78540E+00	1
2	.75000E+00	.37500E+00	-.78540E+00	1
3	.75000E+00	-.37500E+00	-.78540E+00	1
4	.75000E+00	-.11250E+01	.78540E+00	1

	Original axes		Principal axes	
Joint	X-coordinate	Y-coordinate	X-coordinate	Y-coordinate
1	.00000E+00	.00000E+00	-.33333E+02	-.25000E+02
2	.50000E+02	.50000E+02	.16667E+02	.25000E+02
3	.50000E+02	.00000E+00	.16667E+02	-.25000E+02
4	.00000E+00	.50000E+02	-.33333E+02	.25000E+02
Centroid	.3.3333E+02	.25000E+02	.00000E+00	.00000E+00
Principal pole	.71381E+02	.25000E+02	.38048E+02	.71844E-05

Element	Joint-I	Joint-J	Length
1	2	4	.500E+02
2	1	3	.500E+02
3	3	2	.500E+02

Table 2 Continued

Sectional quantities (in relation to principal axes and principal pole)		
F	=	.11901E+05
IXX	=	.311116E+07
IYY	=	.544522E+07
IWW	=	.140146E+10
SE	=	.00000E+00
IXE	=	-.758923E+03
IYE	=	.151785E+04
IWE	=	-.126487E+05
IEE	=	.174837E+05
Angle (in radians) of principal axes	=	.000000E+00

Joint	Wrapping function
1	-.71548E+03
2	-.53452E+03
3	.53452E+03
4	.71548E+03

In the Table 3, for the same cross-section, the sectional quantities, for different laminate stacking sequences in flanges and web, are presented.

Table 3 Variation of sectional properties for different laminate stacking sequences

Laminates lay-up	1 $[0/0]_s$ 2 $[0/0]_s$ 3 $[0/0]_s$	1 $[0/0]_s$ 2 $[0/0]_s$ 3 $[45/-45]_s$	1 $[45/-45]_s$ 2 $[45/-45]_s$ 3 $[0/0]_s$	1 $[45/-45]_2$ 2 $[45/-45]_2$ 3 $[0/0]_s$
F [kN]	.217350E+05	.182200E+05	.147051E+05	.147051E+05
I_{xx} [kNmm^2]	.604293E+07	.487556E+07	.385679E+07	.385679E+07
I_{yy} [kNmm^2]	.105765E+08	.984421E+07	.617751E+07	.617751E+07
$I_{\omega_p\omega_p}$ [kNmm^6]	.272211E+10	.236190E+10	.169708E+10	.169708E+10
S_e [kNmm]	.000000E+00	.000000E+00	.000000E+00	.101190E+04
I_{xe} [kNmm^2]	.000000E+00	-.758923E+03	.000000E+00	-.124637E+05
I_{ye} [kNmm^2]	.000000E+00	.000000E+00	.151785E+04	.000000E+00
$I_{\omega_p e}$ [kNmm^3]	.000000E+00	.000000E+00	-.665521E+05	.000000E+00
I_{ee} [kNmm^2]	.120960E+05	.138919E+05	.156878E+05	.156878E+05
d_C [mm]	.33333E+02	.30118E+02	.37317E+02	.37317E+02
d_P [mm]	.71381E+02	.72971E+02	.68846E+02	.68846E+02

6. Conclusions

The hand calculations of material-geometric properties of a thin-walled composite beam with a complex cross-section are tedious and difficult. The presented computer program provides an opportunity for an automatic evaluation of open al properties of thin-walled composite beams with arbitrary lamination. The geometrical data which need to be entered to perform the analysis have been brought to the minimum: the coordinates of joints, the elements connecting them, thickness and position of layers. The listing of the computer program is given.

Acknowledgments

The present work has been supported by The Ministry of Education and Science of the Republic of Serbia (Project No. ON174027).

References

Banerjee, J.R. (1998), “Free vibration of axially loaded composite Timoshenko beams using the dynamic stiffness matrix method”, *Comput. Struct*., **69**(2), 197-208.

Banerjee, J.R. and Su, H. (2006), “Dynamic stiffness formulation and free vibration analysis of a spinning composite beam”, *Comput. Struct*., **84**(19-20), 1208-1214.

Cardoso, B.J. and Valido, A.J. (2011), “Optimal design of composite laminated thin-walled beams”, *Comput. Struct*., **89**(11-12), 1069-1076

Cardoso, B.J., Benedito, N.M. and Valido, A.J. (2009), “Finite element analysis of thin-walled composite laminated beams with geometrically nonlinear behavior including warping deformation”, *Thin-Wall. Struct*., **47**(11-12), 1363-1372.

Chen, H.H. and Hsiao, K.M. (2007), “Coupled axial–torsional vibration of thin-walled Z-section beam induced by boundary conditions”, *Thin-Wall. Struct.,* **45**(6), 573-583.

Jones, R.M. (1975), *Mechanics of Composite Materials*, Hemisphere Publishing Corporation, New York.

Kim, N.I., Shin, D.K. and Kim, M.Y. (2007), “Improved flexural-torsional stability analysis of thin-walled composite beam and exact stiffness matrix”, *Int. J. Mech. Sci*., **49**(8), 950-969.

Kim, N.I., Shin, D.K. and Park, Y.S. (2008), “Dynamic stiffness matrix of thin-walled composite I beam with symmetric and arbitrary laminations”, *J. Sound Vib*., **318**(1-2), 364-388.

Lee, J. (2001), “Center of gravity and shear center of thin-walled open-section composite beams”, *Compos. Struct*., **52**(2), 255-260.

Machado, S.P. and Cortınez, V.H. (2005), “Non-linear model for stability of thin-walled composite beams with shear deformation”, *Thin-Wall. Struct*., **43**(10), 1615-1645.

Machado, S.P., Filipich, C.P. and Cortinez, V.H. (2007), “Parametric vibration of thin-walled composite beams with shear deformation”, *J. Sound Vib*., **305**(4-5), 563-581.

Murray, N. (1984), *Introduction to the Theory of Thin-Walled Structures*, Claredon Press.

Piovan, M.T. and Cortınez, V.H. (2007), “Mechanics of shear deformable thin-walled beams made of composite materials”, *Thin-Wall. Struct*., **45**(1), 37-62.

Prokic, A. (1996), “New warping function for thin-walled beams. II: Finite element method and applications”, *J. Struct. Eng*., **122**(12), 1443-1453.

Prokić, A. (1999), “Computer program for determination of geometrical properties of thin-walled beams with open profile”, *Adv. Eng. Software*, **30**(2), 109-119.

Prokić, A. (2000), “Computer program for determination of geometrical properties of thin-walled beams with open-closed section”, *Comput. Struct*., **74**(6), 705-715.

Rajasekaran, S. (2005), “Mechanical properties of thin-walled composite beams of generic open and closed

sections", *Struct*. *Eng*. *Mech*., *Int*. *J*., **21**(5), 591-620.
Rajasekaran, S. (2005), "Optimal laminate sequence of non-prismatic thin-walled composite spatial members of generic section", *Compos*. *Struct*., **70**(2), 200-211.
Sapountzakis, E.J. and Mokos, V.G. (2007), "3-D beam element of composite cross section including warping and shear deformation effects", *Comput*. *Struct*., **85**(1-2), 102-116.
Sapountzakis, E.J. and Tsiatas, G.C. (2007), "Flexural–torsional buckling and vibration analysis of composite beams", *Comput*. *Mater*. *Con*., **6**(2), 103-115.
Vo, T.P. and Lee, J. (2009), "On sixfold coupled buckling of thin-walled composite beams", *Compos*. *Struct*., **90**(3), 295-303.
Vo, T.P., Lee, J., Lee, K. and Ahn, N. (2011), "Vibration analysis of thin-walled composite beam with I-shaped cross-sections", *Compos*. *Struct*., **93**(2), 812-820.

CC

Appendix A. Listing of the source program

```
PROGRAM COMPOSITE
C
C      UFILE=NAME OF FILE WITH INPUT DATA
C      JK=DISPLAY OF OUTPUT DATA (3 = PRINTER, 4 = MONITOR)
C      NUMJ = NUMBER OF JOINTS
C      NUME = NUMBER OF ELEMENTS IN CROSS-SECTION
C      X(I) = X-COORDINATE OF JOINT ‘I’
C      Y(I) = Y-COORDINATE OF JOINT ‘I’
C      N1(I) = FIRST JOINT NUMBER OF ELEMENT ‘I’
C      N2(I) = SECOND JOINT NUMBER OF ELEMENT ‘I’
C      TL = THICKNESS OF THE LAYER
C      EL = DISTANCE TO THE CENTROID OF LAYER
C      OL = ANGLE OF ORIENTATION OF LAYER
C      TM = TYPE OF MATERIAL
C
     IMPLICIT REAL*8 (A-H,O-Z)
     INTEGER TM
     CHARACTER*20 UFILE
     COMMON NI,NJ,X2,X1,Y2,Y1,D,S,C
     REAL*8 IXX,IYY,IXY,IXXC,IYYC,IXYC,IWX,IWY,IWXC,IWYC,I1,I2,IWW
    1,IXE,IYE,IEE,IXEC,IYEC,IWE,IE1,IE2
     DIMENSION XN(20),YN(20),W(20),NUMAJ(20)
    1,NEJI(5,20),IACTE(20),N1(20),N2(20),X(20)
    2,Y(20),DD(20),NUMAJA(20),E1(20),E2(20),P12(20),G12(20)
    3,A11(20),B11(20),D11(20),B16(20),D16(20),D66(20)
     WRITE (*,500)
 500 FORMAT (‘ FILE WITH INPUT DATA = ‘)
     READ (*,510) UFILE
 510 FORMAT (A20)
     WRITE (*,511)
 511 FORMAT (‘ OUTPUT DATA (MONITOR = 4,FILE "OUTPUT" = 5)
    1 =‘)
     READ (*,515) JK
 515 FORMAT (I3)
     OPEN (2,FILE=UFILE)
     OPEN (4,FILE=‘CON’,STATUS=‘NEW’)
     OPEN (5,FILE=‘OUTPUT.FOR’)
     READ (2,520) NUMJ,NUME,NUMM
 520 FORMAT (3I5)
     WRITE (JK,525) NUMJ,NUME,NUMM
 525 FORMAT (/,4X,’NUMBER OF JOINTS    =‘,I3,/,4X,’NUMBER OF ELEMENTS
    1  =‘,I3,/,4X,’NUMBER OF MATERIALS  =‘,I3,//)
     WRITE (JK,524)
 524 FORMAT (1X,’TYPE OF MATERIAL’,8X,’E1’,13X,’E2’,13X,’NI’,14X,’G’)
     DO 5 I = 1,NUMM
     READ (2,526) E1(I),E2(I),P12(I),G12(I)
 526 FORMAT (4F10.0)
   5 WRITE (JK,527) I,E1(I),E2(I),P12(I),G12(I)
 527 FORMAT (7X,I3,9X,E12.5,3X,E12.5,3X,E12.5,3X,E12.5)
     DO 10 I = 1,NUMJ
  10 READ (2,530) X(I),Y(I)
 530 FORMAT (2F10.0)
     DO 20 I = 1,NUME
     READ (2,535) N1(I),N2(I),NL
```

```
535 FORMAT (3I5)
    WRITE (JK,528) I
528 FORMAT (//,1X,'ELEMENT',I3,/)
    WRITE (JK,529)
529 FORMAT (1X,'LAMINA',3X,'THICKNESS',2X,'DIST. TO THE CENT. LINE',
   13X,'ANGLE OF ORIENT.',3X,'TYPE OF MATER.')
    DO 20 J = 1,NL
    READ (2,536) TL,EL,OL,TM
536 FORMAT (3F10.0,I5)
    WRITE (JK,541) J,TL,EL,OL,TM
541 FORMAT (I5,2X,E12.5,7X,E12.5,9X,E12.5,10X,I3)
    Q1=E1(TM)/(1-P12(TM)**2*E2(TM)/E1(TM))
    Q2=E2(TM)/(1-P12(TM)**2*E2(TM)/E1(TM))
    Q3=G12(TM)
    Q4=P12(TM)*E2(TM)/(1-P12(TM)**2*E2(TM)/E1(TM))
    CC=COS(OL)
    SS=SIN(OL)
    Q11=Q1*CC**4+Q2*SS**4+2.*CC**2*SS**2*(Q4+2.*Q3)
    Q12=CC**2*SS**2*(Q1+Q2-4.*Q3)+(CC**4+SS**4)*Q4
    Q16=(Q1*CC**2-Q2*SS**2-(Q4+2.*Q3)*(CC**2-SS**2))*CC*SS
    Q22=Q1*SS**4+Q2*CC**4+2.*CC**2*SS**2*(Q4+2.*Q3)
    Q26=(Q1*SS**2-Q2*CC**2+(Q4+2.*Q3)*(CC**2-SS**2))*CC*SS
    Q66=(Q1+Q2-2.*Q4)*CC**2*SS**2+Q3*(CC**2-SS**2)**2
    QQ11=Q11-Q12**2/Q22
    QQ16=Q16-Q12*Q26/Q22
    QQ66=Q66-Q26**2/Q22
    A11(I)=A11(I)+QQ11*TL
    B11(I)=B11(I)+QQ11*EL*TL
    D11(I)=D11(I)+QQ11*(EL**2*TL+TL**3/12.)
    B16(I)=B16(I)+QQ16*EL*TL
    D16(I)=D16(I)+QQ16*(EL**2*TL+TL**3/12.)
 20 D66(I)=D66(I)+QQ66*(EL**2*TL+TL**3/12.)
    DO 30 I = 1,NUME
 30 IACTE(I) = 1
    DO 40 I = 1,NUMJ
    K = 0
    DO 50 J = 1,NUME
    IF ((N1(J).EQ.I).OR.(N2(J).EQ.I)) THEN
    K = K + 1
    NEJI(K,I) = J
    END IF
 50 CONTINUE
    NUMAJ(I) = K
    NUMAJA(I) = K
    IF (NUMAJA(I).GT.2) NUMAJA(I) = 2
 40 CONTINUE
    DO 60 I = 1,NUME
    DD(I) = DSQRT((X(N2(I))-X(N1(I)))**2 + (Y(N2(I))-Y(N1(I)))**2)
    CALL COM (N1(I),N2(I),X(N2(I)),X(N1(I)),Y(N2(I)),Y(N1(I)),DD(I))
    F=F+A11(I)*D
    SX = SX+0.5*A11(I)*(X1+X2)*D+B11(I)*C*D
    SY = SY+0.5*A11(I)*(Y1+Y2)*D+B11(I)*S*D
    IXX = IXX+1./3.*A11(I)*(X1**2+X2**2+X1*X2)*D+B11(I)*(X1+X2)*C*D+
   1D11(I)*C**2*D
    IYY = IYY+1./3.*A11(I)*(Y1**2+Y2**2+Y1*Y2)*D+B11(I)*(Y1+Y2)*S*D+
   1D11(I)*S**2*D
    IXY = IXY+1./6.*A11(I)*(2*X1*Y1+2*X2*Y2+X1*Y2+X2*Y1)*D+
```

```
    10.5*B11(I)*((X1+X2)*S+(Y1+Y2)*C)*D+D11(I)*S*C*D
     SE = SE+2.*B16(I)*D
     IXE = IXE+B16(I)*(X1+X2)*D+2.*D16(I)*C*D
     IYE = IYE+B16(I)*(Y1+Y2)*D+2.*D16(I)*S*D
  60 IEE = IEE+4*D66(I)*D
     K = 0
     II = 0
  90 LL = 0
     DO 70 I = 1,NUMJ
     IF (K.NE.0) II = NUMAJ(I)
     IF ((NUMAJA(I).EQ.1).AND.(II.NE.1)) THEN
     LL = 1
     K = K + 1
     DO 80 J = 1,NUMAJ(I)
     NN = NEJI(J,I)
     IF (IACTE(NN).EQ.1) THEN
     IF (N2(NN).EQ.I) THEN
     KK = N1(NN)
     N1(NN) = N2(NN)
     N2(NN) = KK
     B11(NN) = -B11(NN)
     D16(NN) = -D16(NN)
     END IF
     CALL COM (N1(NN),N2(NN),X(N2(NN)),X(N1(NN)),Y(N2(NN)),Y(N1(NN))
    1,DD(NN))
     W(NJ) = W(NI) + (X1*C + Y1*S)*D
     IACTE(NN) = 0
     NUMAJA(NI) = NUMAJA(NI)-1
     NUMAJA(NJ) = NUMAJA(NJ)-1
     END IF
  80 CONTINUE
     END IF
  70 CONTINUE
     IF (LL.EQ.1) GOTO 90
     DO 110 I = 1,NUME
     CALL COM (N1(I),N2(I),X(N2(I)),X(N1(I)),Y(N2(I)),Y(N1(I)),DD(I))
     HPI=X1*S-Y1*C
     SW = SW+A11(I)*(W(NI)+W(NJ))/2.*D+B11(I)*(HPI-0.5*D)*D
     IWX = IWX+A11(I)*(2.*X1*W(NI)+2.*X2*W(NJ)+X1*W(NJ)+X2*W(NI))/6.*D
    1+B11(I)*((W(NI)+W(NJ))/2.*C-(X1+2.*X2)/6.*D+(X1+X2)/2.*HPI)*D
    2+D11(I)*(HPI*C-0.5*D*C)*D
     IWY = IWY+A11(I)*(2.*Y1*W(NI)+2.*Y2*W(NJ)+Y1*W(NJ)+Y2*W(NI))/6.*D
    1+B11(I)*((W(NI)+W(NJ))/2.*S-(Y1+2.*Y2)/6.*D+(Y1+Y2)/2.*HPI)*D
    2+D11(I)*(HPI*S-0.5*D*S)*D
 110 CONTINUE
     XC = SX/F
     YC = SY/F
     IXXC = IXX-XC**2*F
     IYYC = IYY-YC**2*F
     IXYC = IXY-XC*YC*F
     IXEC=IXE-XC*SE
     IYEC=IYE-YC*SE
     PSI = ATAN(2.*IXYC/(IXXC-IYYC))/2.
     WO = SW/F
     IWXC = IWX-XC*SW
     IWYC = IWY-YC*SW
     I1 = 0.5*(IXXC + IYYC) - 0.5*DSQRT((IYYC-IXXC)**2 + 4.*IXYC**2)
```

```
    I2 = 0.5*(IXXC + IYYC) + 0.5*DSQRT((IYYC-IXXC)**2 + 4.*IXYC**2)
    IE1=IXEC*COS(PSI)+IYEC*SIN(PSI)
    IE2=IYEC*COS(PSI)-IXEC*SIN(PSI)
    XP = (IWYC*IXXC-IWXC*IXYC)/(IXXC*IYYC-IXYC**2)
    YP = (IWYC*IXYC-IWXC*IYYC)/(IXXC*IYYC-IXYC**2)
    XPP = (XP-XC)*COS(PSI) + (YP-YC)*SIN(PSI)
    YPP = -(XP-XC)*SIN(PSI) + (YP-YC)*COS(PSI)
    WRITE (JK,540)
540 FORMAT (/,27X,'ORIGINAL AXES',16X,'PRINCIPAL AXES',/,10X,'JOINT',
   12(5X,'X-COORDINATE',1X,'Y-COORDINATE'),/)
    DO 120 I = 1,NUMJ
    XIC = X(I)-XC
    YIC = Y(I)-YC
    W(I) = W(I)-WO + YP*XIC-XP*YIC
    XN(I) = XIC*COS(PSI) + YIC*SIN(PSI)
    YN(I) = -XIC*SIN(PSI) + YIC*COS(PSI)
120 WRITE (JK,545) I,X(I),Y(I),XN(I),YN(I)
545 FORMAT (11X,I2,6X,E12.5,1X,E12.5,5X,E12.5,1X,E12.5)
    WRITE (JK,550) XC,YC
550 FORMAT(8X,'CENTROID',3X,E12.5,1X,E12.5,7X,'.00000E+00',3X,
   1'.00000E+00')
    WRITE (JK,555) XP,YP,XPP,YPP
555 FORMAT (5X,'PRINCIPAL POLE',E12.5,1X,E12.5,5X,E12.5,1X,E12.5)
    WRITE (JK,560)
560 FORMAT (//,8X,'ELEMENT',4X,'JOINT-I',2X,'JOINT-J',4X,'LENGTH')
    DO 130 I = 1,NUME
    CALL COM (N1(I),N2(I),XN(N2(I)),XN(N1(I)),YN(N2(I)),YN(N1(I))
   1,DD(I))
    HPI=(X1-XPP)*S-(Y1-YPP)*C
    WRITE (JK,565) I,NI,NJ,D
565 FORMAT (10X,I2,8X,I3,6X,I3,4X,E10.3)
    IWW = IWW + A11(I)*(W(NI)**2 + W(NJ)**2 + W(NI)*W(NJ))/3.*D+
   1B11(I)*((W(NI)+W(NJ))*HPI-(W(NI)+2.*W(NJ))/3.*D)*D+
   2D11(I)*(HPI**2-HPI*D+D**2/3.)*D
130 IWE=IWE+B16(I)*(W(NI)+W(NJ))*D+2.*D16(I)*(HPI-0.5*D)*D
    WRITE (JK,571)
571 FORMAT (//,4X,'SECTIONAL QUANTITIES (IN RELATION TO PRINCIPAL AXES
   1 AND PRINCIPAL POLE)')
    WRITE (JK,570) F,I1,I2,IWW,SE,IE1,IE2,IWE,IEE,PSI
570 FORMAT (/,12X,'F                                         =',E12.6
   1,/,12X,'IXX                                  =',E12.6
   2,/,12X,'IYY                                  =',E12.6
   3,/,12X,'IWW                                  =',E12.6
   4,/,12X,'SE                                   =',E12.6
   5,/,12X,'IXE                                  =',E12.6
   6,/,12X,'IYE                                  =',E12.6
   7,/,12X,'IWE                                  =',E12.6
   8,/,12X,'IEE                                  =',E12.6
   9,/,12X,'ANGLE (IN RADIANS) OF PRINCIPAL AXES     =',E12.6)
    WRITE (JK,575)
575 FORMAT (//,8X,'JOINT',5X,'WARPING FUNCTION',/)
    DO 140 I = 1,NUMJ
140 WRITE (JK,580) I,W(I)
580 FORMAT (9X,I2,9X,E12.5)
    END
    SUBROUTINE COM (N1,N2,XNJ,XNI,YNJ,YNI,DD)
    IMPLICIT REAL*8 (A-H,O-Z)
```

```
COMMON NI,NJ,X2,X1,Y2,Y1,D,S,C
NI = N1
NJ = N2
X2 = XNJ
X1 = XNI
Y2 = YNJ
Y1 = YNI
D = DD
S = (X1-X2)/D
C = (Y2-Y1)/D
END
```